121001, 230011, 231011, 380001 WS 2008/2009

Prüfung:	Mathematik 1	Zeit: 90 min
Hilfsmittel:	keine programmierbaren Rechengeräte	
Aufgaben:	beliebig viele [VU,TLM: 18 Punkte = 1,0 7 Punkte = 4,0] [PL: 7 Punkte = bestanden]	

Aufgabe	1	2	3	4	5	6	7	8
Punkte	3	3	3	3	4	3	3	2

Aufgabe 1 Seien $\vec{a} = (2,3,-2)$ *und* $\vec{b} = (1,1,u)$. Bestimmen Sie u so, dass $\vec{a}$ *und* $\vec{b}$ den Winkel 45° einschließen.

Aufgabe 2 $z^4 - 5z^3 - 6z^2 + 70z - 100 = 0$ hat die Lösung $z_1 = 3 - j$. Bestimmen Sie sämtliche Lösungen $z \in \mathbb{C}$ dieser Gleichung.

Aufgabe 3 Bestimmen Sie sämtliche reellen Lösungen von

$$\cos(2x) + \sin x = 0$$

Aufgabe 4 Ermitteln Sie $\lim\limits_{x \to 2\pi} \left[(2\pi - x) \cdot \tan\frac{x}{4} \right]$.

Aufgabe 5 Zeichnen Sie die beiden Bilder von $y = \frac{1}{2}(x + \frac{\pi}{2})$ und $y = \cos x$ für $x \in \left[\frac{-3\pi}{2}, \frac{\pi}{2} \right]$ in ein gemeinsames Schaubild (L.E. 1 cm).

Ermitteln Sie dann näherunsweise sämtliche reellen Lösungen der Gleichung $\frac{1}{2}(x + \frac{\pi}{2}) = \cos x$ (mind. 10^{-4} genau)

-> Seite 2

Aufgabe 6 Bestimmen Sie X aus der Matrizengleichung

$$X \cdot \begin{pmatrix} 1 & 1 & 0 \\ 0 & 1 & 1 \\ 2 & 1 & 3 \end{pmatrix} = \begin{pmatrix} 2 & 1 & 0 \\ 0 & 1 & 1 \\ 1 & 1 & 0 \end{pmatrix}.$$

Aufgabe 7 Ermitteln Sie das Taylorpolynom vom Grad n=4 von

$f(x) = x \cdot \sin x$ an der Entwicklungsstelle $x_0 = \pi$.

Aufgabe 8 Haben die folgenden Gleichungen reelle Lösungen ?

Wenn ja, geben Sie die Lösungen mit Genauigkeit 10^{-4} an.

a) $2\sinh(x) = 5$

b) $2\tanh(x) = 5$

c) $2\cosh(x) = 5$

Hochschule Heilbronn
Technik • Wirtschaft • Informatik
Heilbronn University

121001, 230011, 231011, 380001 WS 2008/2009

Prüfung:	Mathematik 1	Zeit: 90 min
Hilfsmittel:	keine programmierbaren Rechengeräte	
Aufgaben:	beliebig viele [VU,TLM: 18 Punkte = 1,0 7 Punkte = 4,0] [PL: 7 Punkte = bestanden]	

Aufgabe	1	2	3	4	5	6	7	8
Punkte	3	3	3	3	4	3	3	2

Aufgabe 1 Seien $\vec{a}=(2,3,-2)$ *und* $\vec{b}=(1,1,u)$. Bestimmen Sie u so, dass $\vec{a}$ *und* $\vec{b}$ den Winkel 45° einschließen.

```
reset():
P1:=matrix([2, 3, -2]);
P2:=matrix([1, 1, u]);
```

$$\begin{pmatrix} 2 \\ 3 \\ -2 \end{pmatrix}$$

$$\begin{pmatrix} 1 \\ 1 \\ u \end{pmatrix}$$

```
assume(u,Type::Real):
SP:=linalg::scalarProduct(P1,P2);
B1:=norm(P1,2);
B2:=norm(P2,2);
```

$$5-2\cdot u$$

$$\sqrt{17}$$

$$\sqrt{u^2+2}$$

```
glg:=SP/(B1*B2)=cos(PI/4);
```

$$-\frac{\sqrt{17}\cdot(2\cdot u-5)}{17\cdot\sqrt{u^2+2}}=\frac{\sqrt{2}}{2}$$

```
loes:=solve(glg,u);
float(%);
```

$$\left\{-\frac{4\cdot\sqrt{2}\cdot\sqrt{17}}{9}-\frac{20}{9},\ \frac{4\cdot\sqrt{2}\cdot\sqrt{17}}{9}-\frac{20}{9}\right\}$$

$$\{-4.813756398,\ 0.3693119533\}$$

```
prob1:=simplify(subs(lhs(glg)-rhs(glg),u=loes[1]));
prob2:=simplify(subs(lhs(glg)-rhs(glg),u=loes[2]));
```

$$0$$

$$0$$

Aufgabe 2 $z^4 - 5z^3 - 6z^2 + 70z - 100 = 0$ hat die Lösung $z_1 = 3 - j$.

Bestimmen Sie sämtliche Lösungen $z \in \mathbb{C}$ dieser Gleichung.

```
reset():
```

mit 3-j ist auch 3+j Lösung der Gleichung

```
alias(j=I):
p:=z^4-5*z^3-6*z^2+70*z-100;
p1:=expand((z-(3-j))*(z-(3+j)));
```

$$z^4 - 5 \cdot z^3 - 6 \cdot z^2 + 70 \cdot z - 100$$

$$z^2 - 6 \cdot z + 10$$

```
p2:=divide(p,p1);
```

$$z^2 + z - 10, \ 0$$

```
rl:=solve(p2[1],z);
`dies sind die reellen Lösungen`;
```

$$\left\{ -\frac{\sqrt{41}}{2} - \frac{1}{2}, \frac{\sqrt{41}}{2} - \frac{1}{2} \right\}$$

dies sind die reellen Lösungen

```
`Lösungen sind`, {rl[1],rl[2],3+j,3-j};
```

$$\text{Lösungen sind}, \left\{ j + 3, \ 3 - j, \ -\frac{\sqrt{41}}{2} - \frac{1}{2}, \frac{\sqrt{41}}{2} - \frac{1}{2} \right\}$$

Aufgabe 3 Bestimmen Sie sämtliche reellen Lösungen von

$$\cos(2x) + \sin x = 0$$

```
reset():
glg:=cos(2*x)+sin(x)=0;
glg:=subs(glg,cos(2*x)=1-2*sin(x)^2);
```

$$\sin(x)+\cos(2\cdot x)=0$$

$$-2\cdot\sin(x)^2+\sin(x)+1=0$$

loes:=solve(glg,x):
loes:=float(%);

$$\{6.283185307\cdot k+5.759586532 \mid k\in\mathbb{Z}\}\cup\{6.283185307\cdot k+1.570796327 \mid k\in\mathbb{Z}\}\cup\{6.283185307\cdot k+3.665191429 \mid k\in\mathbb{Z}\}$$

Aufgabe 4 Ermitteln Sie $\lim\limits_{x\to 2\pi}\left[(2\pi-x)\cdot\tan\frac{x}{4}\right]$.

reset():
f:=(2*PI-x);
g:=tan(x/4);

$$2\cdot\pi-x$$

$$\tan\left(\frac{x}{4}\right)$$

mupad direkt:

F:=f*g;
hold(limit(f*g,x=2*PI))=limit(F,x=2*PI);

$$-\tan\left(\frac{x}{4}\right)\cdot(x-2\cdot\pi)$$

$$\lim_{x\to 2\cdot\pi} f\cdot g=4$$

ausführlicher

f1:=f;
g1:=1/g;
hold(limit(f/(1/g),x=2*PI))

$$2\cdot\pi-x$$

$$\frac{1}{\tan(\frac{x}{4})}$$

$$\lim_{x\to 2\cdot\pi}\frac{f}{\frac{1}{g}}$$

Zähler- und Nennerableitungen:
F:=diff(f1,x);

```
G:=expand(subs(diff(g1,x),tan(x/4)=sin(x/4)/cos(x/4)));
```

$$-1$$

$$-\frac{\cos(\frac{x}{4})^2}{4\cdot\sin(\frac{x}{4})^2}-\frac{1}{4}$$

```
F/G;
hold(limit(f*g,x=2*PI))=limit(F/G,x=2*PI);
```

$$\frac{1}{\frac{\cos\left(\frac{x}{4}\right)^2}{4\cdot\sin\left(\frac{x}{4}\right)^2}+\frac{1}{4}}$$

$$\lim_{x\to 2\cdot\pi} f\cdot g = 4$$

Aufgabe 5 Zeichnen Sie die beiden Bilder von $y=\frac{1}{2}(x+\frac{\pi}{2})$ und $y = \cos x$ für $x \in \left[\frac{-3\pi}{2},\frac{\pi}{2}\right]$ in ein gemeinsames Schaubild (L.E. 1 cm).

Ermitteln Sie dann näherunsweise sämtliche reellen Lösungen der Gleichung $\frac{1}{2}(x+\frac{\pi}{2}) = \cos x$ (mind. 10^{-4} genau)

```
reset():
f:=cos(x);
g:=1/2*(x+PI/2)
```

$$\cos(x)$$

$$\frac{\pi}{4}+\frac{x}{2}$$

Bild und Rohwert:

```
plo1:=plot::Function2d(f,x=-3*PI/2..PI/2,y=-1..1,Color=RGB::Red):
plo2:=plot::Function2d(g,x=-3*PI/2..PI/2,y=-1..1,Color=RGB::Blue):

plot(plo1,plo2,ViewingBoxYRange=-1..1,Scaling=Constrained);
```



eine erste exakte Lösung liegt bei x1= - Pi /2, die zweite Lösung liegt bei etwa x2 ~ 0,35, die dritte Lösung
liegt dann symmetrisch zu x1, also bei x3= -Pi-x2

```
x0:=0.35;
F:=f-g;
fab:=diff(F,x);
```

$$0.35$$

$$\cos(x) - \frac{x}{2} - \frac{\pi}{4}$$

$$-\sin(x) - \frac{1}{2}$$

```
x0:=0.35;
x1:=float(x0-subs(F,x=x0)/subs(fab,x=x0));
x2:=float(x1-subs(F,x=x1)/subs(fab,x=x1));
```

$$0.35$$

$$0.3250557537$$

$$0.3246980143$$

```
`Lösungen sind`, (-PI/2,x2, float(-PI-x2));
```

Lösungen sind, $-\frac{\pi}{2}$, 0.3246980143, -3.466290668

Aufgabe 6 Bestimmen Sie X aus der Matrizengleichung

$$X \cdot \begin{pmatrix} 1 & 1 & 0 \\ 0 & 1 & 1 \\ 2 & 1 & 3 \end{pmatrix} = \begin{pmatrix} 2 & 1 & 0 \\ 0 & 1 & 1 \\ 1 & 1 & 0 \end{pmatrix}.$$

```
reset():
A:=matrix([[1, 1, 0], [0, 1, 1], [2, 1, 3]]);
B:=matrix([[2, 1, 0], [0, 1, 1], [1, 1, 0]]);
```

$$\begin{pmatrix} 1 & 1 & 0 \\ 0 & 1 & 1 \\ 2 & 1 & 3 \end{pmatrix}$$

$$\begin{pmatrix} 2 & 1 & 0 \\ 0 & 1 & 1 \\ 1 & 1 & 0 \end{pmatrix}$$

```
`Inverse`=1/A;
```

$$\text{Inverse} = \begin{pmatrix} \frac{1}{2} & -\frac{3}{4} & \frac{1}{4} \\ \frac{1}{2} & \frac{3}{4} & -\frac{1}{4} \\ -\frac{1}{2} & \frac{1}{4} & \frac{1}{4} \end{pmatrix}$$

schrittweise:

```
meinMatrixinvers:=proc(Ma)

begin
Pref::matrixSeparator("    "):
ndim:=linalg::nrows(Ma);

G:=matrix(ndim,2*ndim):

for j from 1 to ndim do G[j,ndim+j]:=1 end_for:

for j from 1 to ndim do for i from 1 to ndim do
             G[j,i]:=Ma[j,i]; end_for; end_for;

for j from 1 to (ndim-1) do

k:=0:
for m from j to ndim do
   if (k=0 and G[m,j]<>0) then
   k:=m: G:=linalg::swapRow(G,m,j): end_if;     end_for;
```

```
  for i from (j+1) to ndim do
     G:=linalg::addRow(G,j,i,-G[i,j]/G[j,j]);  end_for;
     print(Typeset,G); end_for;

for j from -ndim to -2   do
     for i from (j+1) to -1 do
     G:=linalg::addRow(G,-j,-i,-G[-i,-j]/G[-j,-j]);  end_for;
     print(Typeset,G);
                                                                    end_for;

for j from 1 to ndim do
 G:=linalg::multRow(G,j,1/G[j,j]): end_for;
 print(Typeset,G);
 Matrix_invers:=linalg::delCol(G,1..ndim);

end_proc;
```

proc meinMatrixinvers(Ma) ... end

```
meinMatrixinvers(A);
```

$$\begin{pmatrix} 1 & 1 & 0 & 1 & 0 & 0 \\ 0 & 1 & 1 & 0 & 1 & 0 \\ 0 & -1 & 3 & -2 & 0 & 1 \end{pmatrix}$$

$$\begin{pmatrix} 1 & 1 & 0 & 1 & 0 & 0 \\ 0 & 1 & 1 & 0 & 1 & 0 \\ 0 & 0 & 4 & -2 & 1 & 1 \end{pmatrix}$$

$$\begin{pmatrix} 1 & 1 & 0 & 1 & 0 & 0 \\ 0 & 1 & 0 & \frac{1}{2} & \frac{3}{4} & -\frac{1}{4} \\ 0 & 0 & 4 & -2 & 1 & 1 \end{pmatrix}$$

$$\begin{pmatrix} 1 & 0 & 0 & \frac{1}{2} & -\frac{3}{4} & \frac{1}{4} \\ 0 & 1 & 0 & \frac{1}{2} & \frac{3}{4} & -\frac{1}{4} \\ 0 & 0 & 4 & -2 & 1 & 1 \end{pmatrix}$$

$$\begin{pmatrix} 1 & 0 & 0 & \frac{1}{2} & -\frac{3}{4} & \frac{1}{4} \\ 0 & 1 & 0 & \frac{1}{2} & \frac{3}{4} & -\frac{1}{4} \\ 0 & 0 & 1 & -\frac{1}{2} & \frac{1}{4} & \frac{1}{4} \end{pmatrix}$$

$$\begin{pmatrix} \frac{1}{2} & -\frac{3}{4} & \frac{1}{4} \\ \frac{1}{2} & \frac{3}{4} & -\frac{1}{4} \\ -\frac{1}{2} & \frac{1}{4} & \frac{1}{4} \end{pmatrix}$$

```
X:=B*1/A;
`ist die gesuchte Matrix X`;
```

$$\begin{pmatrix} \frac{3}{2} & -\frac{3}{4} & \frac{1}{4} \\ 0 & 1 & 0 \\ 1 & 0 & 0 \end{pmatrix}$$

ist die gesuchte Matrix X

Aufgabe 7 Ermitteln Sie das Taylorpolynom vom Grad n=4 von $f(x) = x \cdot \sin x$ an der Entwicklungsstelle $x_0 = \pi$.

```
reset():
f:=x->x*sin(x);
n:=4;
x0:=PI;
```

$x \rightarrow x \cdot \sin(x)$

4

π

```
p:=hold(sum((D@@k)(f)(x0)/k!*(x-x0)^k,k=0..n));
```

$$\sum_{k=0}^{n} \frac{(D@@k)(f)(\mathrm{x0})}{k!} \cdot (x-\mathrm{x0})^k$$

```
p:=eval(p);
```

$$\frac{\pi \cdot (x-\pi)^3}{6} - (x-\pi)^2 + \frac{(x-\pi)^4}{6} - \pi \cdot (x-\pi)$$

Bild (war nicht verlangt):

```
plotfunc2d(f(x),p,x=1..5);
```



Aufgabe 8 Haben die folgenden Gleichungen reelle Lösungen ?

Wenn ja, geben Sie die Lösungen mit Genauigkeit 10^{-4} an.

a) $2\sinh(x) = 5$

b) $2\tanh(x) = 5$

c) $2\cosh(x) = 5$

```
reset():
```

a)

```
x:=float(arcsinh(5/2));
```

1.647231146

b)

es gibt kein reelles x, da $-1 < \tanh(x) < 1$

c)

$\cosh(x) = \cosh(-x)$, cosh ist eine gerade Funktion !

```
x1:=float(arccosh(5/2));
x2:=-float(arccosh(5/2));
```

1.566799237

– 1.566799237